
I wanted to post this solution since we run into this problem all the time: 

Often times, our .bat scripts include making registry changes or copying files which require Administrative access. This 
code will cause the .bat file to trigger a UAC prompt before running AsAdministrator -- allowing us to make those registry 
changes instead of getting permission errors. 

Be careful, though - make sure you map network drives or make environment changes at the end of the file; relaunching a 
file as Administrator means you will not have the same Environment or Map Drives as the User in Windows 8. 

Click here to download the code, below: elevated.bat.zip 

Anyone writing programs or scripts would benefit from reading the MSDN post titled "Teach Your Apps To Play Nicely 
With Windows Vista User Account Control". 

 

@ECHO OFF 

REM Find out what OS we're running... 

REM 5.0=Win2k, 5.1=XP, 5.2=2003, 6.0=Vista, 6=1.Win7, 6.2=Win8, 6.3=Win8.1 

ver | findstr /i "5\.0\." > nul 

IF %ERRORLEVEL% EQU 0 SET MYOS=XP 

ver | findstr /i "5\.1\." > nul 

IF %ERRORLEVEL% EQU 0 SET MYOS=XP 

ver | findstr /i "5\.2\." > nul 

IF %ERRORLEVEL% EQU 0 SET MYOS=VISTA 

ver | findstr /i "6\.0\." > nul 

IF %ERRORLEVEL% EQU 0 SET MYOS=VISTA 

ver | findstr /i "6\.1\." > nul 

IF %ERRORLEVEL% EQU 0 SET MYOS=VISTA 

ver | findstr /i "6\.2\." > nul 

IF %ERRORLEVEL% EQU 0 SET MYOS=VISTA 

ver | findstr /i "6\.3\." > nul 

IF %ERRORLEVEL% EQU 0 SET MYOS=VISTA 

REM Find out what architecture we're on 

IF EXIST C:\Windows\SysWOW64 SET ARCH=64 

IF NOT EXIST C:\Windows\SysWOW64 SET ARCH=32 

ECHO. 

ECHO This script will trigger the UAC prompt on WinVista+ machines. 

REM If not an administrator, relaunch as Admin (generates UAC prompt) 

:checkPrivileges  

NET FILE 1>NUL 2>NUL 

if '%errorlevel%' == '0' ( goto gotPrivileges ) else ( goto getPrivileges )  

:getPrivileges  

if '%1'=='ELEV' (shift & goto gotPrivileges)  

ECHO. 

ECHO. 

ECHO RELAUNCHING WITH ADMIN ACCESS 

ECHO. 

setlocal DisableDelayedExpansion 

set "batchPath=%~0" 

setlocal EnableDelayedExpansion 

ECHO Set UAC = CreateObject^("Shell.Application"^) > "%temp%\OEgetPrivileges.vbs"  

ECHO UAC.ShellExecute "!batchPath!", "ELEV", "", "runas", 1 >> "%temp%\OEgetPrivileges.vbs"  

"%temp%\OEgetPrivileges.vbs"  

exit /B 

:gotPrivileges  

ECHO. 

ECHO Commands after this line are run AsAdministrator, allowing you 

ECHO to copy files into "protected" places. 

ECHO. 

ECHO. 

ECHO OS Version: %MYOS% 

ECHO Architecture: %ARCH%-bit 

ECHO Home: %USERPROFILE% 

ECHO. 

ECHO. 

PAUSE 

